COMPUTER MODELS FOR TROPICAL AND MIDLATITUDE ATMOSPHERES

Ulisses Côrtes Oliveira
Instituto de Aeronáutica e Espaço — ASE-E/IAE/CTA
12228-904 — São José dos Campos–SP, Brazil
E-mail: ulisses@iae.cta.br

Abstract. *The transatmospheric flight of an aerospace vehicle is strongly affected by the atmosphere, which could be modelled with different levels of detail and fidelity. For conceptual and preliminary design of space vehicles and for many other aerospace applications, atmosphere models representative of the mean annual conditions are still essential. Two different models for the atmosphere below* $80\ \mathrm{km}$ *are briefly analyzed here. The first is the International Standard Atmosphere (ISA), which is particularly adequate to midlatitudes, although it has been used for other zones of globe. The second is the International Tropical Reference Atmosphere (ITRA-1986), suitable for the whole of the tropical regions in both the Northern and Southern Hemispheres. The basic equations, an algorithm and the computer codes for these two models are also given. In addition, a comparison among basic properties evaluated with these two models is made with the purpose of illustrating the necessity of using the ITRA-1986 or a regional model for a tropical zone.*

Keywords: *Standard Atmosphere, Reference Atmosphere, International Standard Atmosphere, International Tropical Reference Atmosphere, Atmospheric Properties.*

1. Introduction

The transatmospheric flight of an aerospace vehicle is strongly affected by the prevailing atmospheric conditions. In particular, sounding rockets, missiles, launch and reentry vehicles fly either fully or partially in the atmosphere which influences their trajectory and its dispersion.

The atmosphere mass density is very important since the aerodynamic forces such as the lift and drag are proportional to the dynamic pressure which in turn is proportional to the density. In addition to density, pressure and temperature are also determining parameters to the aerothermal environment experienced by an aerospace vehicle. The adoption of regional standards for the atmosphere has therefore great relevance to its design and performance evolution, flight testing and operation under realistic conditions.

There are detailed models for the earth atmosphere which permit estimate the atmosphere properties characteristics typical of a season, a month, a day or even a local time, at any particular location over the globe. Notwithstanding, a standard atmosphere representative of the mean annual conditions is still essential for many aerospace applications, specially for conceptual design of aerospace vehicles.

The original International Standard Atmosphere (ISA) specified up to $32\ \mathrm{km}$ (NASA and USAF, 1962; ICAO, 1964), and its proposed extension to higher altitudes such as U.S. Standard Atmosphere–1976 (COESA, 1976), have been widely used for meeting these needs. These models were based on conditions prevailing in the temperate region around mid-northern latitudes. However, conditions over the tropics can be substantially different from those specified in the usual International Standards. Therefore, it was necessary to define a standard atmosphere which is close to the mean conditions over tropical zones. The International Tropical Reference Atmosphere (ITRA-1986) proposed by Ananthasaynam and Narasimha (1987) could fill this gap and could be useful for many aerospace applications.

The aim of this article is to present computer routines for ISA and ITRA-1986 models. It commences with a brief historic review of these two models. Then, the basic equations and an algorithm for calculation of atmosphere properties of a generic model are given. Finally, the computer codes (in Pascal) for and a comparison of these models are also presented.

2. Historical Review of the Atmosphere Models

This section gives a historical review of the ISA and ITRA-1986. For this last model, it is also given a more detailed discussion of the tropical characteristics driving the bases and the convenience of the model.

2.1. ISA History

The International Standard Atmosphere, defined in (ISO, 1975), and the U.S. Standard Atmosphere–1976, defined in (COESA, 1976), are the culmination of many years' work and represent advanced stages in a continuing process of defining standard atmospheres. Standard atmospheres were originally developed in the 1920's in the United States and in Europe to satisfy a need for standardization of aircraft instruments and aircraft performance.

The U.S. atmosphere was generated by the National Advisory Committee on Aeronautics (NACA), as documented in Gregg (1922) and later supplements, while the European atmosphere was generated by the International Commission for Aerial Navigation (ICAN), (ICAN, 1924). There were slight differences between these two independently derived atmospheres.

In the post War period a series of meetings with the International Civil Aviation Organization (ICAO) lead to a compromise between the ICAN and the U.S. Standard Atmosphere, the latter using the Toussaint's temperature profile from sea-level all the way up to $20\ \mathrm{km}$, and the former agreeing for sea-level pressure and acceleration due to gravity at $45°\mathrm{N}$ and its altitude dependance. The differences were then reconciled and international uniformity was achieved through adoption by the ICAO in 1952 of a new International Standard Atmosphere (ISA) for altitudes up to 20 km (ICAO, 1954). Work on extending this atmosphere to an altitude of 300 km was undertaken by the U.S. Committee on Extension of the Standard Atmosphere (COESA) and reported in (Minzner et al., 1958). The gathering of more data by rockets and satellites enabled a further extension to 700 km to be made by COESA, which was published in (COESA, 1962). This atmosphere was adopted in 1964 as a new standard by ICAO for altitudes up to 32 km (ICAO, 1964), superseding the 1952 ICAO atmosphere. The U.S. Standard Atmosphere–1976 (COESA, 1976) is a revision and extension to 1000 km of (COESA, 1962) as a result of more experimental data being gathered. It is identical with (COESA, 1962) up to an altitude of 51 km and therefore still agrees with the ICAO standard atmosphere.

In 1975, the International Organization for Standardization (ISO) adopted a standard atmosphere (ISO, 1975), which covers heights up to 80 km. For heights below 50 km the atmosphere is termed "International Standard Atmosphere" while for heights between 50 and 80 km it is termed "Interim Standard Atmosphere". For all practical purposes, the ISO atmosphere is identical with the U.S. Standard Atmosphere–1976. The World Meteorological Organization Standard Atmosphere, defined between -2 km and 32 km, is identical with the data in (ISO, 1975). The most recent extension of ICAO atmosphere is up to $80\ \mathrm{km}'$ in 1994 based on U.S. Standard Atmosphere of 1976 (COESA, 1976).

The current ISA is identical to the U.S. Standard Atmosphere-1976 for geopotential altitudes below $80\ \mathrm{km}'$. It consists of seven successive layers, within which the temperature varies linearly. The base altitudes are at 0, 11, 20, 32, 47, 51, 71 and $80\ \mathrm{km}'$, and the corresponding temperatures in Celsius degrees are 15, -56.5, -56.5, -44.5, -2.5, -2.5, -58.5 and -76.5. The sea-level temperature adopted by ISA ($15°\mathrm{C}$) is generally in the neighborhood of the annual mean in Western cities.

2.2. ITRA History and Importance

It has been well known for centuries that atmospheric conditions in the tropics are different from those at higher latitudes. According to the dictionary, the tropics are the region between $23°28'\mathrm{N}$ and $23°28'\mathrm{S}$, which mark the Tropic of Cancer and Capricorn, respectively. However, it has been pointed out that there can be no sharp dividing line between the tropics and extra tropics, and dynamical considerations suggest $30°\mathrm{N}$ and $30°\mathrm{S}$ as the approximate boundaries of the tropical zone. In true, during summer, tropical conditions prevail up to about 35°N/S, and during winter the change from extratropical to tropical conditions occurs somewhere between 27°N/S and 35°N/S and probably around 30°N/S (Krishna, 1952).

These differences are not confined to the surface or to low altitudes. Evidences came from the work of Ramanathan (1929) which pointed out that a break in the temperature distribution occurs around $16\ \mathrm{km}$ at low latitudes, whereas it occurs at much lower altitudes (around $11\ \mathrm{km}$) in the temperate zone. He also showed that the coldest air over the earth is in the form of a flat ring with temperature about $185\ \mathrm{K}$ at an altitude of approximately $17\ \mathrm{km}$ over the equator; thus, while mean temperatures are higher at sea-level in the tropics, they are lower at altitudes at around $17\ \mathrm{km}$. This behavior, originally observed by Ramanathan (1929), is illustrated in Fig. 1, which is based on more recent data for the month of January (zonal average for the years 1979–1998).

The presence of hemispheric asymmetry in the structure of the middle atmosphere has been broadly confirmed by experimental data (Koshelkov, 1985). Therefore the middle atmosphere of the Southern Hemisphere cannot be regarded as identical to that of the Northern Hemisphere, but as a region with its own specific features of circulation and structure. This fact makes necessary to compile separate reference atmospheres for two hemispheres. However, the insufficiency of experimental data for this purpose has postponed the development of necessary models. Most of the studies on meteorology of the middle atmosphere of the Southern Hemisphere were confined to regional analysis.

In spite of the existing hemispheric asymmetry, the following facts suggest that, with minor modification, it should be possible to provide an Reference Atmosphere suitable for the whole of the tropical regions in both the Northern and Southern Hemispheres. Firstly, studies have shown that, up to $20\ \mathrm{km}$ of altitude, longitudinal variations of atmospheric properties are very weak (Ananthasaynam and Narasimha, 1993). Further, even at altitudes up to $80\ \mathrm{km}$, Cole and Kantor (1978) showed that longitudinal variations during summer are small at all latitudes and at all altitudes above $20\ \mathrm{km}$; during winter longitudinal variations become important only in arctic and subarctic latitudes. Finally, it is well known (Cole and Kantor, 1978) that latitudinal variations are weaker in the tropics than in the temperate regions; hence it should be possible to formulate a meaningful global standard for the tropics.



Figure 1: January 1979–1998 zonally averaged temperature as a function of height.

Tropical mean, of the kind used for the midlatitudes, has been formulated by Pisharoty in 1959. He suggested two standard atmospheres for the tropics, covering altitudes up to 20 km. One of these, called SAAT (Standard Atmosphere for the Asian Tropics), was based on the IMD data for the annual average of the monthly mean temperatures prevailing over four Indian stations (Trivandrum $8°30'$N, Port Blair $11°40'$N, Madras $30°$N and Vishakapatnam $17°43'$N), all around $12°$N. The second standard, called SATU (Standard Atmosphere for the Tropics, Universal), to eliminate possible monsoonal effect was based on the average temperature in the belt $20°$S to $20°$N around the world, based on Goldie's data and showed certain slight differences from SAAT. These differences turned out to be not true based on later measurements.

The recent data for the tropics shows weak longitudinal variations, and tropical conditions probably prevail up to about $30°\mathrm{N/S}$ all the year, as previously discussed. Based on these evidences, it indeed turned out to be possible to provide an International Tropical Reference Atmosphere representative of the whole of the tropical region in both the northern and southern hemispheres (Ananthasaynam and Narasimha, 1987). The proposal of Ananthasaynam and Narasimha (1987), so called ITRA-1986, is also consistent with the averages based on the mean monthly reference atmospheres for various latitudes proposed by Cole and Kantor (1978) and Koshelkov (1985) for the northern and southern hemisphere, respectively, and further based on the Nimbus satellite data of Barnett and Corney (1985) for the global annual average temperature from sea-level up to $80\ \mathrm{km}$.

With regard to the mean sea-level pressure, its variations increase with latitude, the mean value being lower during summer and higher during winter. The value chosen for ITRA-1986 was $1010\ \mathrm{mb}$ (Ananthasaynam and Narasimha, 1985). This value was based on extrapolation of the surface pressure at a large number of Indian stations to the sea-level and a study of about 200 station data in both the northern and southern hemisphere tropical regions. It is also consistent with the data of Cole and Kantor (1978).

The value for the acceleration due to gravity is chosen corresponding to the Tropic of Cancer from the Lambert's formula provided in List (1968), truncated to five decimal places as $9.78852\ \mathrm{m/s^2}$.

The ITRA-1986 proposal (see Tab. 2) consists of linear segments in the temperature distribution with value in Celsius degrees of 27, -9, -74, -5, -5, -74 and -77.6 at geopotential altitudes of 0, 6, 16, 46, 51, 74 and $80\ \mathrm{km}'$, respectively, beyond which the description is in terms of geometric altitude. As discussed before, values for sea-level pressure of 101000 Pa and acceleration due to gravity of $9.78852\ \mathrm{m/s^2}$ are used in the model.

The subsequent section presents the basic equations of a atmosphere model, which will be next grouped in a calculation algorithm that, in turn, will be codded according to the ISA and ITRA-1986 models.

3. Description of the Atmosphere and its Properties

In the lower atmosphere, from sea-level to about $80\ \mathrm{km}$, the air constituents are completely mixed and their relative concentrations are almost constant. These considerations permit to specify the variation of the mean temperature in a simple way, usually in terms of linear segments. Then with a suitable value of the sea-level pressure and temperature distribution the density, the speed of sound and all other transport quantities of interest can be estimated.

Before developing a such simple model it is necessary to show how the the acceleration of gravity and the geopotential altitude can both be derived from the geometric altitude.

3.1. Acceleration of Gravity

By neglecting centrifugal acceleretion, the inverse-square law of gravitation provides an expression for the acceleration due to gravity (g) as a function of geometric altitude z with sufficient accuracy for most model-atmosphere computations:

$$g(z) = \frac{g_o\, R_o^2}{(R_o + z)^2}\ . \tag{1}$$

Here $g_o = g_o(\varphi)$ is the sea-level acceleration of gravity at latitude φ, which is estimated by the Lambert's equation (List, 1968),

$$g_o(\varphi) = 9.78035\ \left(1 + 0.0052885 \sin^2\varphi - 0.0000059 \sin^2 2\varphi\right)\ , \tag{2}$$

and $R_o = R_o(\varphi)$ is the effective radius of the earth that brings harmony between $g_o(\varphi)$ and the vertical gradient of g, assuming that the earth is represented by the International Ellipsoid,

$$R_o(\varphi) = [2g_o(\varphi)] \,/\, \left(3.085462 \times 10^{-6} + 2.27 \times 10^{-9} \cos 2\varphi - 2 \times 10^{-12} \cos 4\varphi\right)\ . \tag{3}$$

3.2. Relationships for Geopotential and Geometric Altitudes

The operational definition of the geopotential altitude is

$$\mathcal{H} = \frac{1}{g_o'} \int_o^z g\, dz\ , \tag{4}$$

where g_o' is a constant numerically equal to the g_o value.

Substituting for $g(z)$, from Eq. (1), in the equation for $\mathcal{H}$, the integration and simplification leads to

$$\mathcal{H} = \Gamma\, \frac{R_o\, z}{R_o + z}\ , \tag{5}$$

or

$$z = \frac{R_o \mathcal{H}}{\Gamma\, R_o - \mathcal{H}}\ , \tag{6}$$

with $\Gamma \equiv g_o/g_o' = 1\ \mathrm{m'/m}$.

3.3. Basic Properties from Sea-Level up to 80 km

Some assumptions should be made in order to derive the properties of a standard, reference or model atmosphere. For altitudes of at least up to $80\ \mathrm{km}$ the usual ones are that the air is dry, free of currents and is thoroughly mixed and thus has a constant molecular mass and obeys the perfect gas law.

The basic equation describing the hydrostatic equilibrium along the vertical direction is

$$\frac{dp}{dz} = -\rho\, g\ , \tag{7}$$

where p is the pressure, z is the geometric altitude, ρ is the air density, and g is the acceleration due to gravity.

Equation (4) can be rewritten as

$$g_o'\, d\mathcal{H} = g\, dz\ . \tag{8}$$

By using Eq. (8), the Eq. (4) could be set in terms of the geopotential altitude: the Eq. (7) becomes

$$\frac{dp}{d\mathcal{H}} = -\rho\, g_o'\ . \tag{9}$$

The perfect gas law is

$$p = \frac{\rho \mathcal{R} T}{\mathcal{M}} \,, \tag{10}$$

where $\mathcal{R}$ is the universal constant, $\mathcal{M}$ is the mean molecular mass of air, and T its temperature. Eliminating ρ between the two previous equations, and integrating in p, one gets the variation of pressure p, in terms of $\mathcal{H}$, as

$$p(\mathcal{H}) = p_\mathrm{o} \, \exp\left\{ \frac{g_\mathrm{o}' \mathcal{M}}{\mathcal{R}} \int_0^{\mathcal{H}} \frac{d\mathcal{H}}{T(\mathcal{H})} \right\} , \tag{11}$$

p_o being the sea-level ($\mathcal{H} = 0$) pressure. If the temperature in the various altitude segments is defined by[1]

$$T = T_\mathrm{M} = T_\mathrm{M,b} + L_\mathrm{M,b}(\mathcal{H} - \mathcal{H}_\mathrm{b}) \,, \tag{12}$$

then the pressure in the corresponding levels is given by

$$p = p_\mathrm{b} \left[\frac{T_\mathrm{M,b}}{T_\mathrm{M}(\mathcal{H})} \right]^{\left(\frac{g_\mathrm{o}' \mathcal{M}_\mathrm{o}}{\mathcal{R}\, L_\mathrm{M,b}} \right)} , \qquad \text{if } L_\mathrm{M,b} \neq 0, \tag{13}$$

or

$$p = p_\mathrm{b} \, \exp\left[- \frac{g_\mathrm{o}' \, \mathcal{M}_\mathrm{o} (\mathcal{H} - \mathcal{H}_\mathrm{b})}{\mathcal{R} \, T_\mathrm{M,b}} \right] , \qquad \text{if } L_\mathrm{M,b} = 0 \,. \tag{14}$$

The mass density could be easily calculated from the equation of perfect gases:

$$\rho = \frac{\mathcal{M}}{\mathcal{R}} \frac{p}{T} \,. \tag{15}$$

The speed of sound is given by

$$a = \sqrt{\gamma \frac{p}{\rho}} = \sqrt{\gamma \frac{\mathcal{R}}{\mathcal{M}} T} \,, \tag{16}$$

where $\gamma = 1.4$ is the ratio of specific heats of air.

In the previous formulae, T_M is the *molecular-scale temperature*, which is related to the *kinetic temperature* (T) by $T_\mathrm{M}/T = \mathcal{M}_\mathrm{o}/\mathcal{M}$, where $\mathcal{M}_\mathrm{o}$ calls for the molecular mass of the air at the sea-level altitude. Note that $T = T_\mathrm{M}$ for altitudes below $80\ \mathrm{km}$, since $\mathcal{M} = \mathcal{M}_\mathrm{o}$ in this range. The molecular-scale temperature varies linearly between the geopotential altitudes at which it is specified, according to Eq. (12). $T_\mathrm{M,b}$ denotes the molecular-scale temperature at the base altitudes $\mathcal{H}_\mathrm{b}$, and $L_\mathrm{M,b}$ is the gradient of the molecular-scale temperature with geopotential altitude (angular coefficient for the linear approximation). The index $b = 0$ corresponds to the sea-level ($\mathcal{H} = 0$), for which the temperature and the pressure are both known. The values of T_o, p_o and $L_\mathrm{M,b}$ ($b = 0, \ldots, n$) are model-dependent. This way, the calculation of atmospheric properties for a given model can be done sequentially for any number of linear temperature segments (n).

3.4. ISA Base Points and Parameters

The ISA can be used in the 10° latitude band around 45°N/S, and it was noticed the average conditions were however reasonably close to the condition at 45°N/S.

Table 1 presents the numerical values of the parameters that define ISA. The essential ones, from which the others can be determined, are shown in red.

3.5. ITRA Base Points and Parameters

The ITRA-1986 was proposed to be used in the 30° latitude band around the equator, i.e., from 30°S to 30°N, and represents approximately the annual average conditions prevailing in this band, which are however reasonably close to the condition at 23.5°N/S.

In ITRA-1986, as well as in most models, it is used linear temperature distribution with altitude, between successive base points. This linear distribution makes possible a closed form integration for obtaining the atmospheric properties. Values of parameters at the base altitudes for the ITRA-1986 are given in Tab. 2. Again, the red color indicates the essential ones.

[1] Usually it is used a linear temperature distribution with altitude. This is done since a closed form integration for obtaining the atmospheric properties is then possible.

Table 1: Parameters at the base altitudes for the ISA model.

b	$\mathcal{H}$ (km′)	z (km)	T (K)	L_b (K/km′)	p_b (Pa)
0	0	0.00	288.15	−6.5	101325.00
1	11	11.02	216.65	0.0	22632.06
2	20	20.06	216.65	+1.0	5474.89
3	32	32.16	228.65	+2.8	868.02
4	47	47.35	270.65	0.0	110.91
5	51	51.41	270.65	−2.8	66.94
6	71	71.80	214.65	−2.0	3.96
7	80	81.02	196.65		0.89

Table 2: Parameters at the base altitudes for the ITRA model.

b	$\mathcal{H}$ (km′)	z (km)	T (K)	L_b (K/km′)	p_b (Pa)
0	0	0.00	300.15	−6.0	101000.00
1	6	6.01	264.15	−6.5	48861.38
2	16	16.04	199.15	+2.3	11102.42
3	46	46.34	268.15	0.0	134.87
4	51	51.41	268.15	−3.0	71.41
5	74	74.87	199.15	−0.6	2.43
6	80	81.02	195.55	0.0	0.86

3.6. Reference Values for Each Model

The reference latitude for each model is not an arbitrary value for which the appropriate value of $g_o(\varphi)$ must be found but, rather, for ISA, φ is the latitude associated with the so-called standard acceleration of gravity equal to $9.80665\ \mathrm{m/s^2}$ as designated in all U.S. Standard Atmospheres published after 1922. It corresponds to the latitude 45°32’33", and the associated value of the effective earth radius is $R_o = 6,356,766\ \mathrm{m}$.

For the sea-level acceleration due to gravity in the ITRA-1986 it is used a value corresponding to the Tropic of Cancer (latitude 23°28’N), which from Lambert’s formula, Eq. (2), is $9.78852\ \mathrm{m/s^2}$ (truncated to five decimal places). The associated effective earth radius is $R_o = 6,341,744\ \mathrm{m}$, according to Eq. (3). A sea-level pressure of $1010\ \mathrm{mb}$ has been chosen based on experimental data. This value of the sea-level pressure is somewhat lower than the mid-latitude value of $1013.25\ \mathrm{mb}$ in ISA.

The sea-level values used in ISA and ITRA-1986 models for some interesting quantities are given in Tab. 3.

Table 3: Sea-level values of atmospheric properties in ISA and ITRA-1986.

Quantity	**Symbol**	**Unit**	**ISA Value**	**ITRA Value**
Geopotential altitude	$\mathcal{H}_o$	km′	0	0
Geometric altitude	z_o	km	0	0
Kinetic temperature	T_o	K	288.5	300.15
Pressure	p_o	Pa	101325	101000
Density	ρ_o	$\mathrm{kg/m^3}$	1.225	1.172
Acceleration due to gravity	g_o	$\mathrm{m/s^2}$	9.80665	9.78852
Mean molecular mass	$\mathcal{M}_o$	kg/(kmol)	28.9644	28.9644

3.7. Other Useful Properties

In the preceding sections it was given formulae for estimation of temperature, pressure, mass density and speed of sound. Next, for completeness, formulae for estimation of other useful atmospheric quantities are also presented.

The coefficient of dynamic viscosity could be estimated by *Sutherland’s formula*:

$$\mu = \frac{\beta T^{3/2}}{T + \mathrm{Su}}\,, \tag{17}$$

with $\beta = 1.458 \times 10^{-6}\ \mathrm{kg/(s \cdot m \cdot K^{1/2})}$ and $\mathrm{Su} = 110.4\ \mathrm{K}$ (*Sutherland’s constant*).

The coefficient of kinematic viscosity is

$$\nu = \frac{\mu}{\rho} \, . \tag{18}$$

The coefficient of thermal conductivity is approximated by

$$k = \frac{(2.64638 \times 10^{-3}) \cdot T^{3/2}}{T + \left(245.4 \times 10^{-12/T}\right)} \, . \tag{19}$$

The *mean particle speed*, the *mean free path*, and the *mean collision frequency* are given, respectively, by

$$v = \left[\frac{8\mathcal{R}\,T}{\pi\mathcal{M}}\right]^{1/2} \, , \tag{20}$$

$$\lambda = \frac{\sqrt{2}\mathcal{R}\,T}{2\pi \mathrm{N_A} \sigma^2 p} \, , \tag{21}$$

$$\eta = \frac{v}{\lambda} \, . \tag{22}$$

In the above formulae, $\mathrm{N_A} = 6.022169 \times 10^{26}\ \mathrm{kmol}^{-1}$ is the Avogadro's constant, and $\sigma = 3.65 \times 10^{-10}\ \mathrm{m}$ is the mean effective collision diameter.

3.8. Algorithm for Calculation of Atmospheric Properties

For aerospace applications, in general, it is necessary to determine the atmospheric properties for given value of the geometric altitude z. Utilizing the relation between the geometric and the geopotential altitude, the latter can be calculated, for a given latitude. In the lower atmosphere below $80\ \mathrm{km}$, knowing the sea-level pressure and the temperature description from sea-level to $80\ \mathrm{km}$, the pressure and the density distributions can be worked out. The coefficients of dynamic and kinematic viscosities, the coefficient of thermal conductivity and the speed of sound, which are meaningful in the lower atmosphere, can also be worked out. The mean particle speed, collision frequency, mean free path and the mean collision frequency can be obtained by using the appropriate formulae valid in the lower and upper atmospheres.

The sequence above described could be better understood by putting the proper formulae in an algorithmic form, as shown in Fig. 2, where z and φ are the given parameters.

$$g_\mathrm{o} = 9.78035\ \left(1 + 0.0052885 \sin^2\varphi - 0.0000059 \sin^2 2\varphi\right)$$

$$R_\mathrm{o} = (2g_\mathrm{o}) / \left(3.085462 \times 10^6 + 2.27 \times 10^9 \cos 2\varphi - 2 \times 10^{12} \cos 4\varphi\right)$$

$$\mathcal{H} = \frac{R_\mathrm{o}\, z}{(R_\mathrm{o} + z)}$$

$$T = T_\mathrm{M,b} + L_\mathrm{M,b}(\mathcal{H} - \mathcal{H}_\mathrm{b})$$

$$p = p_\mathrm{b} \left[\frac{T_\mathrm{M,b}}{T}\right]^{\frac{g_o' \mathcal{M}_\mathrm{o}}{\mathcal{R}\, L_\mathrm{M,b}}} \, , \qquad L_\mathrm{M,b} \neq 0$$

$$p = p_\mathrm{b}\ \exp\left[-\frac{g_o\, \mathcal{M}_\mathrm{o} (\mathcal{H} - \mathcal{H}_\mathrm{b})}{\mathcal{R}\, T_\mathrm{M,b}}\right] \, , \qquad L_\mathrm{M,b} = 0$$

$$\rho = \frac{\mathcal{M}_\mathrm{o}}{\mathcal{R}}\ \frac{p}{T}$$

$$a = \sqrt{\gamma \frac{p}{\rho}} = \sqrt{\gamma \frac{\mathcal{R}}{\mathcal{M}_\mathrm{o}} T}$$

$$g(z) = \frac{g_\mathrm{o}\, R_\mathrm{o}^2}{(R_\mathrm{o} + z)^2} \, .$$

Figure 2: Algorithm for calculation of atmosphere models.

3.9. Computer Codes for the Atmosphere Models ITRA and ISA

The International Standard Atmosphere (ISA) and the International Tropical Reference Atmosphere (ITRA-1986) models were analyzed in the preceding sections. The computer codes (in Pascal) for these two models are given next.

They correspond to the algorithm of Fig. 2. However, the values of g_o and R_o for each model were previously set in order to make the codes computationally more efficient.

3.9.1. Pascal Code for ISA

```
procedure Atmos_ISA(Z: real; var P,T,RHO,A: real);
{ Author: Ulisses Cortes Oliveira
  Version date: 10/March/2004
  Calculates the atmospheric properties according to the International
  Standard Atmosphere (ISA) and the U.S. Standard Atmosphere-1976
  (USSA-1976), for geometric altitudes from 0 up to 80 km.
  Input datum: geometric altitude (Z in meters).
  The following variables are passed by reference to the calling routine:
        P   = Atmospheric pressure (Pa)
        T   = Kinetic temperature (K)
        RHO = Mass density (kg/m3)
        A   = Sound speed  (m/s) }
type
  Vec7r = array[0..7] of real;
const
  G = 1.4;              { Air specific heats ratio, G = Cp/Cv }
  MMc = 28.9644e-3; { Air Molecular Mass at the sea-level (kg/mol ) }
  R = 8.31432;          { Perfect Gas constant (N.m/mol.K) }
  Pc = 101325.0;        { Atmospheric pressure at sea-level (Pa) }
  Rc = 6356766;         { Effective Earth radius (adopted) (m) }
  Gc = 9.80665;         { Acceleration of gravity at sea-level (m/s^2) }
  Hb:  Vec7r = (0.0,11.0e3,20.0e3,32.0e3,47.0e3,51.0e3,71.0e3,84852.0);
  Lmb: Vec7r = (-6.5e-3,0.0,1.0e-3,2.8e-3,0.0,-2.8e-3,-2.0e-3,0); {K/m}
  Tmb: Vec7r=(288.15,216.65,216.65,228.65,270.65,270.65,214.65,196.65);{K}
var
  b,i: -1..7;   { Indices and counters }
  Tm,           { "Molecular-Scale" Temperature (K) }
  H,            { Geopotential Altitude (m') }
  DZ: real;     { Altitude Increment (m) }
  Pb: Vec7r;    { Pressure constants }
begin  { procedure Atmos_ISA }
  { Calculation of pressure at the base altitudes }
  Pb[0] := Pc;
  for b := 0 to 6 do
    if Lmb[b] = 0 then
      Pb[b+1] := Pb[b]*Exp(-Gc*MMc*(Hb[b+1] - Hb[b])/(R*Tmb[b]))
    else
      Pb[b+1] := Pb[b]*Exp((Gc*MMc/(R*Lmb[b]))*Ln(Tmb[b]/Tmb[b+1]));
  H := Rc*Z/(Rc + Z);
  b := 7;  { for H >= Hb[7] }
  for i := 0 to 6 do if (H >= Hb[i]) then b := i;
  Tm := Tmb[b] +  Lmb[b]*(H - Hb[b]);  { linear interpolation }
  T := Tm;
  if Lmb[b] = 0 then
    P := Pb[b]*Exp(-Gc*MMc*(H - Hb[b])/(R*Tmb[b]))
  else
    P := Pb[b]*Exp((Gc*MMc/(R*Lmb[b]))*Ln(Tmb[b]/Tm));
  RHO := P*MMc/(R*Tm);
  A := Sqrt(G*R*Tm/MMc);
end;  { procedure Atmos_ISA }
```

3.9.2. Pascal Code for ITRA-1986

```
procedure Atmos_ITRA86(Z: real; var P,T,RHO,A: real);
{ Author: Ulisses Cortes Oliveira
  Version date: 10/March/2004
  Calculates the atmospheric properties according to the
  International Tropical Reference Atmosphere (ITRA-1986),
  for geometric altitudes from 0 up to 80 km.
  Input datum: geometric altitude (Z in meters).
  The following variables are passed by reference to the calling routine:
       P   = Atmospheric pressure (Pa)
       T   = Kinetic temperature (K)
       RHO = Mass density (kg/m3)
       A   = Sound speed  (m/s) }
type
  Vec6r = array[0..6] of real;
const
  G = 1.4;            { Air specific heats ratio, G = Cp/Cv }
  MMc = 28.9644e-3; { Air Molecular Mass at the sea-level (kg/mol ) }
  R = 8.31432;        { Perfect Gas constant (N.m/mol.K) }
  Pc = 101000.0;      { Atmospheric pressure at sea-level (Pa) }
  Rc = 6341744;       { Effective Earth radius (adopted) (m) }
  Gc = 9.78852;       { Acceleration of gravity at sea-level (m/s^2) }
  Hb:  Vec6r = (0, 6e3, 16e3, 46e3, 51e3, 74e3, 80e3);  {geopot. alt., m}
  Lmb: Vec6r = (-6.0e-3, -6.5e-3, 2.3e-3, 0.0, -3.0e-3, -0.6e-3,0); {K/m}
  Tmb: Vec6r = (300.15, 264.15, 199.15, 268.15, 268.15, 199.15,195.55);{K}
var
  b,i: -1..7;    { Indices and counters }
  Tm,            { "Molecular-Scale" Temperature (K) }
  H,             { Geopotential Altitude (m') }
  DZ: real;      { Altitude Increment (m) }
  Pb: Vec6r;     { Pressure constants }
begin  { procedure Atmos_ITRA86 }
  { Calculation of pressure at the base altitudes }
  Pb[0] := Pc;
  for b := 0 to 5 do
    if Lmb[b] = 0 then
      Pb[b+1] := Pb[b]*Exp(-Gc*MMc*(Hb[b+1] - Hb[b])/(R*Tmb[b]))
    else
      Pb[b+1] := Pb[b]*Exp((Gc*MMc/(R*Lmb[b]))*Ln(Tmb[b]/Tmb[b+1]));
  H := Rc*Z/(Rc + Z);
  b := 6;  { for H >= Hb[6] }
  for i := 0 to 6 do if (H >= Hb[i]) then b := i;
  Tm := Tmb[b] + Lmb[b]*(H - Hb[b]); { linear interpolation }
  T := Tm;
  if Lmb[b] = 0 then
    P := Pb[b]*Exp(-Gc*MMc*(H - Hb[b])/(R*Tmb[b]))
  else
    P := Pb[b]*Exp((Gc*MMc/(R*Lmb[b]))*Ln(Tmb[b]/Tm));
  RHO := P*MMc/(R*Tm);
  A := Sqrt(G*R*Tm/MMc);
end; { procedure Atmos_ITRA86 }
```

4. Comparison of the Atmospheric Models

The vertical profiles of temperature, pressure, mass density, and speed of sound in accordance with ISA and ITRA-1986 models are shown in Fig. 3. In addition, Fig. 4 present the variation along the geopotential altitude for the ratios "ITRA-1986/ISA" of these atmospheric properties, for better comparison.



Figure 3: Vertical profiles of some atmospheric properties in ISA and in ITRA-1986.

The inspection for the Figs. 3 and 4 shows that ITRA-1986 model gives greater values of the following properties: mass density, in the geopotential altitude range from about 10 to 78 km'; pressure, in the range 2 to 78 km'; and temperature and speed of sound in the ranges from 0 to 13 km' and 27 to 46 km'. In all the other altitude ranges, the ISA model gives greater values than the ITRA-1986 model. The larger differences between these two models are near the tropopause, since the tropical model is characterized by a colder tropopause, which is not succeeded by an isothermal layer as in the midlatitude model.

Although it is not ease to preview the intensity of the effects of these differences, it is certainly possible to anticipate that they contribute largely to the aerothermal environment of the vehicle, affecting, therefore, its design and performance.

5. Final Remarks

The longitudinal variation of the earth atmosphere is very weak. In the other hand, the variations with the latitude become progressively more sensible for locations more distant from the equator. So, an atmospheric model developed for middle latitudes could not be appropriate for calculations in the tropical zone.

The paper compares atmosphere models considering their applicability to tropical regions. Two different annual average models for the atmosphere were briefly analyzed here. The first is the International Standard Atmosphere (ISA), which is identical to the U.S. Standard Atmosphere-1976 (USSA-1976) below $80\ \mathrm{km}$. It is particularly adequate to midlatitudes, although it has been used for other zones of globe. The second is the International Tropical Reference Atmosphere (ITRA-1986), suitable for the whole of the tropical regions in both the Northern and Southern Hemispheres.

One of the main contributions of this study is to point out differences in atmospheric models that may affect the performance of aerospace vehicles. The basic equations, an algorithm and the computer codes for calculations based on these two models were also given. In addition, a comparison among basic properties evaluated with these two models was made with the purpose of illustrating the necessity of using the ITRA-1986 or a regional model for a tropical zone. It was shown that the ITRA-1986 can provide property estimations sensibly different from those given by the more used ISA model, which is proper for midlatitudes.

The simplicity of the analyzed models makes them particularly attractive for using in early design phases of launch vehicles and sounding rockets. They are applicable to altitudes ranging from the sea-level up to $80\ \mathrm{km}$. For above this limit, the approach suggested by Ananthasaynam and Narasimha (1993) could be adopted with minor updates.



Figure 4: Ratios of the atmospheric properties in ITRA-1986 and ISA at various altitudes.

6. References

Ananthasayanam, M. R. and Narasimha, R., 1985, “A Proposed International Tropical Reference Atmosphere Up To 80 km”, *Handbook for MAP*, No. 16, pp. 197–204, July.

Ananthasayanam, M. R. and Narasimha, R., 1987, “A Proposed International Tropical Reference Atmosphere up to 1000 km”, *Adv. Space Res.*, Vol. 7, No. 10, pp. 117–131.

Ananthasayanam, M. R. and Narasimha, R., 1993, “Reference Annual Atmospheres for Midlatitude and Tropics”, *Adv. Space Res.*, Vol. 13, No. 1, pp. (1)95–(1)104.

Barnett, J. J. and Corney, M., 1985, “Middle Atmosphere Reference Model Derived from Satellite Data”, *Handbook for MAP*, Vol. 16, pp. 47–85.

COESA (U.S. Committee on Extension to the Standard Atmosphere), 1962, *U.S. Standard Atmosphere, 1962*. U.S. Government Printing Office.

COESA (U.S. Committee on Extension to the Standard Atmosphere), 1976, *U.S. Standard Atmosphere, 1976*. U.S. Government Printing Office.

Cole, A. E., and Kantor, A. J., 1978, *Air Force Reference Atmospheres*, AFGL-TR-78-0051.

Gregg, W. R., 1922, *Standard Atmosphere*, NACA Report 147.

ICAN (International Commission for Air Navigation), 1924, “Resolution No. 192”, ICAN Official Bulletin No. 7, pp. 34–35.

ICAO (International Civil Aviation Organization), 1954, *Manual of the ICAO Standard Atmosphere*. Document 7488.

ICAO (International Civil Aviation Organization), 1964, *Manual of the ICAO Standard Atmosphere to 32 km (105,000 ft)*, Doc. 7488/2, Second Edition.

ISO (International
Organization for Standardization), 1975, *Standard Atmosphere*. ISO-2533-1975(E).

Koshelkov, Yu. P., 1985, “Observed Winds and Temperatures in the Southern Hemisphere”, *Handbook for MAP*, Vol. 16.

Krishna, Rao P. R., 1952, *Indian J. Meteorol. Geophys.*, 3, p. 173.

List, R. J., 1968, “Smithsonian Meteorological Tables”, Sixth Revised Edition, Fourth Reprint, Smithsonian Institute, Washington, D.C.

Minzner, R.A. Ripley, W.S. Condron, T.P., 1958, *U.S. Extension to the ICAO Standard Atmosphere Tables and Data to 300 Standard Geopotential kilometers*. Geophys. Res. Dir. and U.S. Weather Bureau.

Minzner, R.A., Champion K.S.W., and Pond A.L., 1959, “The ARDC Model Atmosphere, 1959”, *Air Force Surveys in Geophysics*, No. 115, AFCRL-TR-59-267.

NASA and U.S. Air Force, 1962, *U.S. Standard Atmosphere, 1962*, U.S. Government Printing Offices, Washnigton, DC.

Pisharoty, P. R., 1959, “A Standard Atmosphere for the Tropics”, (msl to $20\ \mathrm{km}$), *Indian Jour. of Met. and Geophysics*, 10, 243.

Ramanathan, K. R., 1929, “Distribution of Temperature in the first 25 kilometers over the Earth.” *Nature*, 123, pp. 834–835.

7. Responsibility Notice

The author is the only responsible for the printed material included in this paper.